



# Introduction to Bruker Pulse Programs



# Where to find the pulseprogram

- Computer:  
`/opt/topspin/exp/stan/nmr/lists/pp/user`  
or: `cd $pp` (changes direct to user folder)
- Topspin: edcpul , edpul
- Display: commandline: spdisp

# Where to find the pulseprogram



# Where to find the pulseprogram

The screenshot displays the Bruker TopSpin 3.2 software interface. The main window is titled "HSQC\_Test\_May2010 1 1 /opt/topspin/data/setup/nmr". The left sidebar shows a file browser with a tree structure of folders and files. The right pane shows the "PulseProg" tab, which displays the pulse program file "hsqc15N.eth" located at "/opt/topspin/exp/stan/nmr/lists/pp/user". A red box highlights the "PulseProg" tab and the file path. A red arrow points from the "PulseProg" tab to the file path. The file path is also highlighted in blue. The file path is: `File: hsqc15N.eth (/opt/topspin/exp/stan/nmr/lists/pp/user)`. The file content is displayed in the main window, showing various pulse program parameters and definitions.

File: hsqc15N.eth (/opt/topspin/exp/stan/nmr/lists/pp/user)

```

;15N-1H HSQC correlations without
;The delay for 3-9-19 watergate
;with 1/d;d=distance of next null

;S. Mori et al, JMR B108, 94-98

;p11 : power for 1H
;p12 : power for 13C
;p13 : power for 15N
;p113 : power for 15N waltz16 de

;p1 : 90 degree hard pulse 1H
;p3 : 90 degree hard pulse 13C
;p4 : 180 degree hard 13C pulse
;p5 : 90 degree hard pulse 15N
;pcpd3 : 90 deg cpd-pulse15N(walt

;d1 : relaxation delay
;d2 : INEPT delay (~2.7m)
;d5 : delay for 3-9-19=1/(Hz
;in0 : 1/(2 SW) (Hz)

;p21 : 500u (Gradient in first
;p22 : 500u (Gradient for z-fil
;p23 : 1m (Gradient for second
;gpz1 : 19%
;gpz2 : 30%
;gpz3 : 65%

;$OWNER=setup
#include <Avance.incl>

"in0=inf1/2"

define delay INEPT_W
define delay INEPT_D

#define GRADIENT1 10u p21:gp1 200u
#define GRADIENT2 10u p22:gp2 200u
#define GRADIENT3 10u p23:gp3 200u

"p2=2*p1"
"p6=2*p5"


"d0=in0/2-p5*2/3-1.4159-p1"
"d3=d5/2-p5"
"INEPT_D=d2-p21-210u"
"INEPT_W=d2-(p23+210u+p1*2.3846+d5*2.5)"



```

# Where to find the pulseprogram


**S** shows the prcompiled pp

← E

 or 'spdisp'



# A very simple 1D



line oriented

➔ each line an action

Starting by zg or gs:

1 executable internal binary form

➔ syntax errors are reported

2 compiled pp is loaded into hardware

➔ measurement begins

```
;zg.eth
```

```
;1D sequence
```

```
ze  
2 d1 pl1:f1  
p1 ph1  
go=2 ph0  
wr #0  
exit
```

```
ph1=0  
ph0=0
```

Text after semicolon = Comment



**ze** zero (NS; memory), applies DS (zd → no DS)

**2** label

**d1** delay

**pl1:f1** powerlevel pl1 on frequency channel f1 (1H)

**p1 ph1** pulse p1 with phase ph1 (default: f1))

**go=2 ph0** data acquisition, loop to 2 (NS-1 times), phase cycling

➔ phase cycling also used during DS

**wr #0** write data to file

➔ data only stored/accessible on disk after all NS scans  
use 'tr' on the command line to store

**exit** end of pp

Phase cycles

# Pulses & Delays

## Pulses

- p0 ... p63
- define pulse p135  
define pulse p30d1H
- manipulate duration:  
p1\*1.5  
p3\*0.33
- calculate pulses  
"p13=p14-d3/2"
- rectangular pulse power
  - plW0 ... plW63
  - set with a delay: 10u p11
- shaped pulses
  - (p1:sp1 ph8):f1  
! 2u p11:f1

## Delays

- d0 ... d63
- define delay d135
- define delay relax
- manipulate duration:  
d2\*1.5  
d3\*0.33
- calculate delays  
"d13=3s+aq-dw\*10"
- 3.5u, 10m, 0.1s → fixed delays
- Incrementing / decrementing delays
  - id1 = d1+ IN[1]
  - dd1 = d1 – IN[1]
  - rd1 resets d1



# Comments & Predefinitions

```
;zg.eth
```

```
;1D sequence
```

```
;d1      : relaxation delay  
;p1      : power for 1H  
;p1      : 90 degree hard pulse 1H
```

format for comments on parameters  
→ displayed in ased

```
;$OWNER=setup
```

Ownership

```
#include <Avance.incl>
```

Includes definitions in the file `Avance.incl`,  
stored in the Bruker pp folder  
→ can add def. in personal file

```
1 ze  
2 d1 p1:f1  
  (p1 ph1):f1  
  go=2 ph0  
  wr #0  
exit
```

```
ph1=0  
ph0=0
```

# z-Gradient

```
;zg.eth
```

```
;1D sequence
```

```
;d1      : relaxation delay
```

```
;p11     : power for 1H
```

```
;p1      : 90 degree hard pulse 1H
```

```
;p21     : 1 ms (Gradient before acquisition)
```

```
;gpz1    : 50 %
```

suggested gradient length/strength

```
;$OWNER=setup
```

```
#include <Avance.incl>
```

```
1 ze
```

```
2 d1
```

```
  p21:gp1
```

```
  10m p11:f1
```

```
  (p1 ph1):f1
```

```
  go=2 ph0
```

```
  wr #0
```

```
exit
```

shaped gradient: defined in file  
(eg. SINE.100)

```
ph1=0
```

```
ph0=0
```

# Gradient and Auto-shimming

```
;zg.eth
```

```
;1D sequence
```

```
;d1      : relaxation delay
```

```
;p11     : power for 1H
```

```
;p1      : 90 degree hard pulse 1H
```

```
;p21     : 1 ms (Gradient before acquisition)
```

```
;gpz1    : 50 %
```

```
;$OWNER=setup
```

```
#include <Avance.incl>
```

```
1 ze      ← LOCKH_OFF
2 d1      ← LOCKH_ON
  p21:gp1
  10m p11:f1
  (p1 ph1):f1
  go=2 ph0
  wr #0
exit      ← LOCKH_OFF
```

LOCKH\_OFF and LOCKH\_ON  
→ defined in Avance.incl

```
ph1=0
ph0=0
```

# Gradient and Auto-shimming

```
;zg.eth
```

```
;1D sequence
```

```
;d1      : relaxation delay
```

```
;p11     : power for 1H
```

```
;p1      : 90 degree hard pulse 1H
```

```
;p21     : 1 ms (Gradient before acquisition)
```

```
;gpz1    : 50 %
```

```
;$OWNER=setup
```

```
#include <Avance.incl>
```

```
1 ze
```

```
2 10u LOCKH_OFF
```

```
d1
```

```
10u LOCKH_ON
```

```
p21:gp1
```

```
10m p11:f1
```

```
(p1 ph1):f1
```

```
go=2 ph0
```

```
wr #0
```

```
10u LOCKH_OFF
```

```
exit
```

```
ph1=0
```

```
ph0=0
```

# Define Placeholder: #define

```
;zg.eth

;1D sequence
;d1      : relaxation delay
;p11     : power for 1H
;p1      : 90 degree hard pulse 1H
;p21     : 1 ms (Gradient before acquisition)
;gpz1    : 50 %
```

```
;$OWNER=setup
```

```
#include <Avance.incl>
```

```
#define GRADIENT1 10u p21:gp1 200u
```

defines placeholder GRADIENT1

```
1 ze
2 10u LOCKH_OFF
  d1
  10u p11:f1
  10u LOCKH_ON
  GRADIENT1
  (p1 ph1):f1
  go=2 ph0
  wr #0
  10u LOCKH_OFF
exit
```

```
ph1=0
ph0=0
```

# Define Parameter: define

```
;zg.eth

;1D sequence
;d1      : relaxation delay
;p11     : power for 1H
;p1      : 90 degree hard pulse 1H
;p21     : 1 ms (Gradient before acquisition)
;gpz1    : 50 %
```

```
;$OWNER=setup
```

```
#include <Avance.incl>
```

```
#define GRADIENT1 10u p21:gp1 200u
```

```
define pulse proton90
"proton90=p1"
```

defines proton90 to be a pulse, length = p1

```
1 ze
2 10u LOCKH_OFF
  d1
  10u p11:f1
  10u LOCKH_ON
  GRADIENT1
  (proton90 ph1):f1
  go=2 ph0
  wr #0
  10u LOCKH_OFF
exit
```

```
ph1=0
ph0=0
```

# Phase Cycling

```
;zg.eth

;1D sequence
;d1      : relaxation delay
;p11     : power for 1H
;p1      : 90 degree hard pulse 1H
;p21     : 1 ms (Gradient before acquisition)
;gpz1    : 50 %

;$OWNER=setup

#include <Avance.incl>

#define GRADIENT1 10u p21:gp1 200u

1 ze
2 10u LOCKH_OFF
  d1
  10u p11:f1
  10u LOCKH_ON
  GRADIENT1
  (p1 ph1):f1
  go=2 ph0
  wr #0
  10u LOCKH_OFF
exit
```

```
ph1=0 1 2 3
ph0=0 1 2 3
```

Phase cycle: select signal / suppress artifacts  
0, 1, 2, 3 → 0°, 90°, 180°, 270° (or x, y, -x, -y)

# Phase Cycling

- ph0 ... ph31
- at the end of the pp
- syntax: ph1= 1 0 0 1 2 3 3 2  
or ph1= 1 0 0 1  
2 3 3 2
- 0 ➔ 0° (x), 1 ➔ 90° (y), 2 ➔ 180° (-x), 3 ➔ 270° (-y)
- next scan next phase
- defined to one channel (p1 ph1):f1
- adding constant to phase (phcor): (p1 ph8:r):f1
- various abbreviations/calculations possible



# Water Suppression: Presaturation

```
;zgpr.eth

;1D sequence with presaturation
;d1      : relaxation delay
;p11     : power for 1H
;p1      : 90 degree hard pulse 1H
;p19     : power level for presaturation
;p21     : 1 ms (Gradient before acquisition)
;gpz1    : 50 %

;$OWNER=setup

#include <Avance.incl>

#define GRADIENT1 10u p21:gp1 200u

1 ze
2 10u p19:f1
  10u LOCKH_OFF
  d1 cw:f1
  10u do:f1
  10u p11:f1
  10u LOCKH_ON
  GRADIENT1
  (p1 ph1):f1
  go=2 ph0
  wr #0
  10u LOCKH_OFF
exit

ph1=0 1 2 3
ph0=0 1 2 3
```

cw → starts "continuous wave" at p19  
do → stops cw

# Water Suppression: Watergate

```
;zg-wg3919.eth
```

```
;1D sequence with watergate using 3-9-19
```

```
;d1      : relaxation delay
```

```
;p11     : power for 1H
```

```
;p1      : 90 degree hard pulse 1H
```

```
;d5      : delay 3-9-19=1/(Hz between nulls)
```

```
;p21     : 1000u (Gradient before and after 3-9-19)
```

```
;gpz1    : 35%
```

```
;$OWNER=setup
```

```
#include <Avance.incl>
```

```
#define GRADIENT1 10u p21:gp1 200u
```

```
1 ze
```

```
2 10u LOCKH_OFF
```

```
   d1 p11:f1
```

```
   10u LOCKH_ON
```

```
   (p1 ph1):f1
```

```
   GRADIENT1
```

```
   (p1*0.2308 ph3 d5 p1*0.6923 ph3 d5 p1*1.4615 ph3):f1
```

```
   d5
```

```
   (p1*1.4615 ph4 d5 p1*0.6923 ph4 d5 p1*0.2308 ph4):f1
```

```
   GRADIENT1
```

```
   go=2 ph0
```

```
   wr #0
```

```
   10u LOCKH_OFF
```

```
exit
```

```
ph1=0 2
```

```
ph0=0 2 2 0
```

```
ph3=1 1 2 2 3 3 0 0
```

```
ph4=3 3 0 0 1 1 2 2
```

3-9-19 Watergate segment

# Heteronuclear Decoupling

```
;zg-wg3919-dec.eth

;1D sequence with watergate using 3-9-19 with decoupling
;d1      : relaxation delay
;p11     : power for 1H
;p1      : 90 degree hard pulse 1H
;pcpd2   : 90 degree for 13C decoupling (~90us)
;p112    : power level for 13C decoupling
;pcpd3   : 90 degree for 15N decoupling (~180us)
;p113    : power level for 15N decoupling
;d5      : delay 3-9-19=1/(Hz between nulls)
;p21     : 1000u (Gradient before and after 3-9-19)
;gpz1    : 35%

;$OWNER=setup

#include <Avance.incl>

#define GRADIENT1 10u p21:gp1 200u

1 ze
2 10u
  10u do:f2
  10u do:f3
  10u LOCKH_OFF
  d1 p11:f1
  10u LOCKH_ON
  10u p112:f2
  10u p113:f3
  (p1 ph1):f1
  GRADIENT1
  (p1*0.2308 ph3 d5 p1*0.6923 ph3 d5 p1*1.4615 ph3):f1
  d5
  (p1*1.4615 ph4 d5 p1*0.6923 ph4 d5 p1*0.2308 ph4):f1
  GRADIENT1
  go=2 ph0 cpd2:f2 cpd3:f3
  wr #0
  10u do:f1
  10u do:f2
  10u do:f3
  10u LOCKH_OFF
exit

ph1=0 2
ph0=0 2 2 0
ph3=1 1 2 2 3 3 0 0
ph4=3 3 0 0 1 1 2 2
```

## composite pulse decoupling

cpds1 ...cpds8 → synchronous  
cpd1 ...cpd8 → asynchronous

```
....
pcpd*3:180
pcpd :0
pcpd*2:180
pcpd*4:0
pcpd*2:180
....
```

do stops cpd on given channel

# Improve Readability

```
;zg-wg3919-dec.eth

;1D sequence with watergate using 3-9-19 with decoupling
;d1      : relaxation delay
;pl1     : power for 1H
;p1      : 90 degree hard pulse 1H
;pcpd2   : 90 degree for 13C decoupling (~90us)
;pl12    : power level for 13C decoupling
;pcpd3   : 90 degree for 15N decoupling (~180us)
;pl13    : power level for 15N decoupling
;d5      : delay 3-9-19=1/(Hz between nulls)
;p21     : 1000u (Gradient before and after 3-9-19)
;gpz1    : 35%

;$OWNER=setup

#include <Avance.incl>

#define GRADIENT1 10u p21:gp1 200u

1 ze
2 10u
10u do:f2
10u do:f3
10u LOCKH_OFF
d1 pl1:f1
10u LOCKH_ON
10u pl12:f2
10u pl13:f3
(p1 ph1):f1
GRADIENT1
(p1*0.2308 ph3 d5 p1*0.6923 ph3 d5 p1*1.4615 ph3):f1
d5
(p1*1.4615 ph4 d5 p1*0.6923 ph4 d5 p1*0.2308 ph4):f1
GRADIENT1
go=2 ph0 cpd2:f2 cpd3:f3
wr #0
10u do:f1
10u do:f2
10u do:f3
10u LOCKH_OFF
exit

ph1=0 2
ph0=0 2 2 0
ph3=1 1 2 2 3 3 0 0
ph4=3 3 0 0 1 1 2 2
```

→ DecouplingOFF

→ watergate3919

→ DecouplingON

# Improve Readability

```
;zg-wg3919-dec.eth
```

```
;1D sequence with watergate using 3-9-19 with decoupling
;d1      : relaxation delay
;pl1     : power for 1H
;p1      : 90 degree hard pulse 1H
;pcpd2   : 90 degree for 13C decoupling (~90us)
;pl12    : power level for 13C decoupling
;pcpd3   : 90 degree for 15N decoupling (~180us)
;pl13    : power level for 15N decoupling
;d5      : delay 3-9-19=1/(Hz between nulls)
;p21     : 1000u (Gradient before and after 3-9-19)
;gpz1    : 35%
```

```
;$OWNER=setup
```

```
#include <Avance.incl>
```

```
#define GRADIENT1 10u p21:gp1 200u
```

```
#define DecouplingOFF 10u do:f2 \n 10u do:f3
```

```
#define DecouplingON cpd2:f2 cpd3:f3
```

```
#define Watergate3919 GRADIENT1 \n (p1*0.2308 ph3 d5 p1*0.6923 ph3 d5 p1*1.4615 ph3):f1 \n d5 \n (p1*1.4615 ph4 d5 p1*0.6923 ph
```

```
1 ze
```

```
2 10u
```

```
DecouplingOFF
```

```
10u LOCKH_OFF
```

```
d1 pl1:f1
```

```
10u LOCKH_ON
```

```
10u pl12:f2
```

```
10u pl13:f3
```

```
(p1 ph1):f1
```

```
Watergate3919
```

```
go=2 ph0 DecouplingON
```

```
wr #0
```

```
10u do:f1
```

```
DecouplingOFF
```

```
10u LOCKH_OFF
```

```
exit
```

```
ph1=0 2
```

```
ph0=0 2 2 0
```

```
ph3=1 1 2 2 3 3 0 0
```

```
ph4=3 3 0 0 1 1 2 2
```

# Improve Readability

```
;zg-wg3919-dec.eth
```

```
;1D sequence with watergate using 3-9-19 with decoupling
;d1      : relaxation delay
;p11     : power for 1H
;p1      : 90 degree hard pulse 1H
;pcpd2   : 90 degree for 13C decoupling (~90us)
;p112    : power level for 13C decoupling
;pcpd3   : 90 degree for 15N decoupling (~180us)
;p113    : power level for 15N decoupling
;d5      : delay 3-9-19=1/(Hz between nulls)
;p21     : 1000u (Gradient before and after 3-9-19)
;gpz1    : 35%
```

```
;$OWNER=setup
```

```
#include <Avance.incl>
```

```
#include 'home/setup/predef.incl'
```

```
1 ze
2 10u
  DecouplingOFF
  10u LOCKH_OFF
  d1 p11:f1
  10u LOCKH_ON
  10u p112:f2
  10u p113:f3
  (p1 ph1):f1
  Watergate3919
  go=2 ph0 DecouplingON
  wr #0
  10u do:f1
  DecouplingOFF
  10u LOCKH_OFF
exit
```

```
ph1=0 2
ph0=0 2 2 0
ph3=1 1 2 2 3 3 0 0
ph4=3 3 0 0 1 1 2 2
```

```
#define GRADIENT1 10u p21:gp1 200u
```

```
#define DecouplingOFF 10u do:f2 \n 10u do:f3
```

```
#define DecouplingON cpd2:f2 cpd3:f3
```

```
#define Watergate3919 GRADIENT1 \n\
(p1*0.2308 ph3 d5 p1*0.6923 ph3 d5 p1*1.4615 ph3):f1 \n\
d5 \n\
(p1*1.4615 ph4 d5 p1*0.6923 ph4 d5 p1*0.2308 ph4):f1 \n\
GRADIENT1
```

# Improve Readability

```
;zg-wg3919-dec.eth
```

```
;1D sequence with watergate using 3-9-19 with decoupling
;d1      : relaxation delay
;p11     : power for 1H
;p1      : 90 degree hard pulse 1H
;pcpd2   : 90 degree for 13C decoupling (~90us)
;p112    : power level for 13C decoupling
;pcpd3   : 90 degree for 15N decoupling (~180us)
;p113    : power level for 15N decoupling
;d5      : delay 3-9-19=1/(Hz between nulls)
;p21     : 1000u (Gradient before and after 3-9-19)
;gpz1    : 35%
```

```
;$OWNER=setup
```

```
#include <Avance.incl>
```

```
#include 'home/setup/predef.incl'
```

```
1 ze
2 10u
  DecouplingOFF
  Set_power
  Relax_delay
  Pulse1H
  Watergate3919
go=2 ph0 DecouplingON
wr #0
10u do:f1
  DecouplingOFF
10u LOCKH_OFF
exit

ph1=0 2
ph0=0 2 2 0
ph3=1 1 2 2 3 3 0 0
ph4=3 3 0 0 1 1 2 2
```

```
#define GRADIENT1 10u p21:gp1 200u
```

```
#define DecouplingOFF 10u do:f2 \n 10u do:f3
```

```
#define DecouplingON cpd2:f2 cpd3:f3
```

```
#define Watergate3919 GRADIENT1 \n\
(p1*0.2308 ph3 d5 p1*0.6923 ph3 d5 p1*1.4615 ph3):f1 \n\
d5 \n\
(p1*1.4615 ph4 d5 p1*0.6923 ph4 d5 p1*0.2308 ph4):f1 \n\
GRADIENT1
```

```
#define Set_power 10u pl12:f2 \n 10u pl13:f3 \n 10u pl1:f1
```

```
#define Relax_delay 10u LOCKH_OFF \n d1 \n\
10u LOCKH_ON
```

```
#define Pulse1H (p1 ph1):f1
```

# Questions ?

**pp manual:**

TopSpin → ? → Manual (docs)

→ Pulse Programming